

Add Line Endings

Ajoute un caractère retour (CR) ou un caractère retour avec saut de ligne (CRLF) à plusieurs lignes de texte, avec la possibilité de contraindre les lignes à une longueur déterminée. Cette commande vous donne le contrôle, avec AppleScript, de la commande "Add CR/LFs (Mac -> DOS)" de Tex-Edit et vous permet de passer outre la préférence de longueur de lignes choisies par défaut.

Cette commande vous permet de traduire un texte Mac en texte DOS, pour le transmettre, par exemple, à un serveur non-Mac ou un BBS.

L'exemple qui suit signifie que, si une fenêtre de document se trouve au premier plan de l'écran, alors il faut ajouter un caractère de fin de ligne de telle manière que chaque ligne ne contienne pas plus de vingt-cinq caractères.

```
tell application "Tex-Edit Plus F"  
  if window 1 exists then tell window 1  
    add line endings appending cr with line length 25  
  end tell  
end tell
```

Ce script ne modifie en rien les options de longueur de lignes choisies par défaut.

Notez que le caractère de retour à la ligne fait lui-même partie des vingt-cinq caractères autorisés pour une ligne. Cela signifie aussi que, si l'on omet le paramètre with line length , alors les options définies par défaut seront utilisées pour tronquer les lignes.

Balance

Introduit dans la version 2.9 de Tex-Edit Plus, la commande Balance vous permet d'utiliser avec AppleScript les fonctionnalités de la commande Balance qui se trouve dans le menu Menu Spécial.

La commande de Menu Balance est surtout utile pour les programmeurs, ceux qui utilisent TE+ pour écrire du code source. Elle étend une sélection de texte à la parenthèse, l'accolade ou le crochet, suivant la sélection, permettant ainsi un bon contrôle des paires de caractères ouvrant et fermant d'un fragment de texte.

```
tell application "Tex-Edit Plus F"  
  balance selection  
end tell
```

Block Format

Ajoute une ligne de texte avant et après un bloc de texte sélectionné et les caractères de texte précédant chaque ligne dans ce bloc, avec la possibilité de contraindre les lignes à une longueur déterminée. Cette commande vous permet de contrôler, avec AppleScript, la commande "**Block Format**" de Tex-Edit, en outrepassant la préférence "Block Options" choisie par défaut.

Cette commande indente une section courte de texte (quoiqu'il n'y a pas de limite de taille, en fait), pour donner au texte un effet "plus dramatique", ainsi que vous pouvez désirer le faire dans un courriel avec l'emploi des citations. En outrepassant les options retenues dans les préférences de formatage de bloc, vous pouvez ainsi créer plusieurs configurations de formatage pour différents textes et caractères que vous voulez utiliser dans un même document, ou automatiser la procédure de formatage de bloc.

Le script s'assure que la sélection contient quelque chose (si cela est, la couleur du texte sélectionné est en vidéo inverse). Si aucune sélection n'existe, alors l'exécution du script s'achève.

```
tell application "Tex-Edit Plus F"  
  if selection is not "" then  
    block format selection -voir "selection (objet)"  
  end if  
end tell
```

Dans ce cas, nous avons utilisé les préférences de formatage bloc utilisées par défaut. Nous pouvons toutefois outrepasser ces choix en ajoutant des paramètres au script.

```
tell application "Tex-Edit Plus F"  
  if selection is not "" then  
    block format selection ↵  
      with line header ↵  
      "-- " with leading line ↵  
      "-- Mother Goose writes:" with trailing line "-- Memo: lamb stays home"  
  end if  
end tell
```

J'utilise dans l'exemple précédent les options considérées par défaut qui donnent une longueur maximum de 60 caractères aux lignes; mais je peux changer cela, dans un nouveau script, en choisissant un nombre compris entre 20 et 200:

```
tell application "Tex-Edit Plus F"  
  if selection is not "" then  
    block format selection ↵  
      with line header ↵  
      "-- " with leading line ↵  
      "-- Mother Goose writes:" with trailing line ↵  
      "-- Memo: lamb stays home" with line length 25  
  end if  
end tell
```

Notez que si nous pouvons outrepasser les options choisies par défaut, ces options ne s'en trouvent pas modifiées pour autant.

Change Case

Modifie la casse d'un texte pour le mettre entièrement en majuscule ou en minuscule, à moins que vous ne désiriez modifier que la casse du premier caractère d'un mot ("word") ou d'une phrase.

Nous recevons parfois, par E-Mail, des textes où la présence immodérée de phrases en majuscule crie à nos yeux comme un son trop fort saurait le faire à nos oreilles. Nous calmons le jeu avec cette commande.

```
tell application "Tex-Edit Plus F"  
  if selection is not "" then  
    change case selection into lowercase  
  end if  
end tell
```

Les cris ont cessés, mais le texte manque dorénavant de capitales en début de phrase. Nous ajouterons donc le paramètre sentence caps :

```
tell application "Tex-Edit Plus F"  
  if selection is not "" then  
    change case selection into sentence caps  
  end if  
end tell
```

Change Size

Cette commande, introduite dans la version 2.7 de Tex-Edit Plus, vous permet d'augmenter ou de diminuer la ou les tailles d'un objet texte spécifié, qui peut être un caractère ("character"), un mot ("word"), une ligne ("line"), un paragraphe ("paragraph"), ou une sélection ("selection (objet)", mais pas une fenêtre ("window") (quoiqu'il existe une astuce pour ce faire).

Ce script augmente d'un point la taille de la police de caractères de la première ligne du document se trouvant au premier plan de l'écran.

```
tell application "Tex-Edit Plus F"  
  change size line 1of window 1 increasing by 1  
end tell
```

Pour revenir à la situation antérieure à l'exécution de ce dernier script, nous allons à présent diminuer la taille de la police de caractères, en utilisant pour cela un nombre négatif de même valeur absolu que le précédent.

```
tell application "Tex-Edit Plus F"  
  change size line 1of window 1 increasing by -1  
end tell
```

Pour modifier la taille de la police de caractères d'un document entier, il suffit d'utiliser contents of comme référence de l'objet.

```
tell application "Tex-Edit Plus F"  
  change size contents of window 1 increasing by 1  
end tell
```

Maintenant, mettons un peu de style à notre texte. Augmentons d'un point la taille du premier caractère de chaque paragraphe.

```
tell application "Tex-Edit Plus F"  
  change size character 1 of word 1 of every paragraph of window 1 increasing by 4  
end tell
```

Close

Ferme la ou les fenêtres spécifiées.

```
tell application "Tex-Edit Plus"  
  close front window  
end tell
```

Copy

Copie la sélection dans le Presse-papiers.

Cet exemple copie une sélection de texte d'un document dans le Presse-papiers, crée un nouveau document, puis colle le contenu du Presse-papiers dans ce nouveau document.

```
tell application "Tex-Edit Plus F"  
  if selection is not "" then  
    copy  
    make new window  
    paste  
  end if  
end tell
```

Notez que si ce dernier script n'offre pas grand intérêt, l'association de la commande Copy avec les commandes de recherche de texte peut rendre cette commande plus attrayante.

Copy Unstyled

Copie la sélection dans le Presse-papiers, en ignorant les attributs de styles.

```
tell application "Tex-Edit Plus F"  
  if selection is not "" then  
    copy unstyled  
    make new window  
    paste  
  end if  
end tell
```

Count

Retourne le nombre d'éléments, d'une classe particulière, se trouvant au sein d'un objet.

Compter le nombre de fenêtres.

```
tell application "Tex-Edit Plus F"  
  count window  
end tell
```

résultat: 2

Compter le nombre de mots d'un document.

```
tell application "Tex-Edit Plus F"  
  tell document 1  
    count word  
  end tell  
end tell
```

résultat: 22

Cut

Copie la sélection dans le Presse-papiers, puis supprime l'objet sélectionné.

```
tell application "Tex-Edit Plus F"  
  cut  
end tell
```

Data Size

Retourne la taille, calculée en octets, d'un objet.

```
tell application "Tex-Edit Plus F"  
  tell document 1  
    data size  
  end tell  
end tell
```

résultat: 284

Delete

Détruit le ou les objets spécifiés.

```
tell application "Tex-Edit Plus F"  
  tell document 1  
    delete word 1  
  end tell  
end tell
```

Duplicate

Duplique un objet.

```
tell application "Tex-Edit Plus F"
```

```
tell document 1
  duplicate word 1
end tell
end tell
```

Exists

Vérifie si un objet existe.

```
tell application "Tex-Edit Plus F"
  tell document 1
    exists word 1
  end tell
end tell
```

résultat: true

Get

Récupère la valeur (ou la référence) d'un ou plusieurs objets spécifiés.

```
tell application "Tex-Edit Plus F"
  tell document 1
    get word 1
  end tell
end tell
```

résultat: "Mary"

Insert Contents

Insère le contenu d'un fichier donné.

Les commandes **Insert**, introduites dans la version 2.7 de Tex-Edit Plus, permettent un contrôle de placement de texte divers, de caractères et de fichiers.

```
set leFichier to choose file with prompt "Choisir un fichier:" of type {"TEXT "} -- voir l'osax
"choose file"
```

```
tell application "Tex-Edit Plus F"
  tell document 1
    insert contents of leFichier at selection
  end tell
end tell
leFichier
```

leFichier: alias "gwen:testInsertion"

Insert Date

Insère la date courante, telle qu'elle est définie sur votre ordinateur, à l'emplacement que vous avez désigné dans un document Tex-Edit.

Cette insertion peut prendre cette forme-ci: lundi 10 juillet 2000

Les commandes **Insert**, introduites dans la version 2.7 de Tex-Edit Plus, permettent un contrôle de placement de texte divers, de caractères et de fichiers.

```
tell application "Tex-Edit Plus F"  
  tell document 1  
    set n to count paragraph  
    set n to n + 1  
    insert date at paragraph n  
  end tell  
end tell
```

```
tell application "Tex-Edit Plus F"  
  insert date at line 1 of window 1  
end tell
```

Insert Page Break

Insère un saut de page, générant une nouvelle page pendant l'impression. Les commandes **Insert**, introduites dans la version 2.7 de Tex-Edit Plus, permettent un contrôle de placement de texte divers, de caractères et de fichiers.

```
tell application "Tex-Edit Plus F"  
  tell document 1  
    insert page break at selection  
  end tell  
end tell
```

Insert Sounds

Insère tous les sons d'un fichier donné. Les commandes **Insert**, introduites dans la version 2.7 de Tex-Edit Plus, permettent un contrôle de placement de textes divers, de caractères et de fichiers.

```
set leFichier to choose file with prompt "Choisir un fichier:" of type {"sfil"} -- voir "choose file"  
tell application "Tex-Edit Plus F"  
  tell document 1  
    insert sounds of leFichier at selection  
  end tell  
end tell  
leFichier
```

```
leFichier: alias "gwen:Laser"
```

Insert Time

Insère l'heure courante, telle qu'elle est définie sur votre ordinateur, à l'emplacement que vous avez désigné dans un document Tex-Edit. Cette insertion peut prendre cette forme-ci: 18:58
Les commandes **Insert**, introduites dans la version 2.7 de Tex-Edit Plus, permettent un contrôle de placement de texte divers, de caractères et de fichiers.

```
tell application "Tex-Edit Plus F"
```

```
tell document 1
  insert time at selection
end tell
end tell
```

Make

Crée un nouvel objet.

```
tell application "Tex-Edit Plus F"
  tell document 1
    make new paragraph with data "nouveau paragraphe"
  end tell
end tell
```

résultat: line 8 of window 1 of application "Tex-Edit Plus F"

Move

Déplace un objet vers un nouvel emplacement.

```
tell application "Tex-Edit Plus F"
  tell document 1
    move paragraph to paragraph 0
  end tell
end tell
```

résultat: text from word 1 to word 2 of window 1 of application "Tex-Edit Plus F"

```
tell application "Tex-Edit Plus F"
  tell document 1
    move paragraph to before paragraph 1
  end tell
end tell
```

résultat: text from word 1 to word 2 of window 1 of application "Tex-Edit Plus F"

Open

Ouvre le ou les fichiers spécifiés.

```
set leFichier to choose file with prompt "Choisir un fichier:" of type {"TEXT "} -- voir "choose file"
tell application "Tex-Edit Plus F"
  open leFichier
end tell
```

résultat: window 1 of application "Tex-Edit Plus F"

Paste

Colle le contenu du Presse-papiers dans la sélection.

```
tell application "Tex-Edit Plus F"
  paste
end tell
```

Paste Style

Applique les attributs de style se trouvant dans le Presse-papiers au contenu de la sélection (voir "selection (objet)").

Il vous faut évidemment avoir du texte dans le Presse-papiers et du texte sélectionné dans un document.

```
tell application "Tex-Edit Plus F"  
  paste style  
end tell
```

Paste Unstyled

Colle le contenu du Presse-papiers dans la sélection, en ignorant les styles.

```
tell application "Tex-Edit Plus F"  
  paste unstyled  
end tell
```

Playback

Joue un son, ou une animation, donnée qui se trouve dans un document.

```
tell application "Tex-Edit Plus F"  
  tell document 1  
    playback selection  
  end tell  
end tell
```

Pour le document affiché ci-dessus en illustration, le script suivant aurait le même effet.

```
tell application "Tex-Edit Plus F"  
  tell document 1  
    playback sound 1  
  end tell  
end tell
```

Print

Imprime le ou les documents spécifiés.

```
tell application "Tex-Edit Plus F"  
  activate  
  print document 1  
end tell
```

Print One Copy

Imprime le ou les documents spécifiés en évitant la fenêtre de dialogue d'impression.

```
tell application "Tex-Edit Plus F"
```

```
print one copy document 1
end tell
```

Print Selection

Imprime le texte sélectionné dans la ou les fenêtres spécifiées.

```
tell application "Tex-Edit Plus F"
  tell document 1
    activate
    print selection
  end tell
end tell
```

Quit

Quitte l'application.

```
tell application "Tex-Edit Plus F"
  quit
end tell
```

Reopen

Réactive une application déjà lancée.

```
tell application "Tex-Edit Plus F"
  reopen
end tell
```

Replace

Remplace toutes les occurrences d'un texte donné par un autre texte. C'est probablement l'une des commandes les plus puissantes de la suite Tex-Edit. Les opérations de remplacement ne porte pas seulement sur le texte pure, il vous permet aussi d'adopter un format particulier pour les caractères, tels que les espaces et les tabulations.

Admettons que nous avons écrit un livre sur George Washington, mais que nous avons commis l'erreur systématique d'écrire Gorge et non George. Ce script corrigera l'erreur:

```
tell application "Tex-Edit Plus F"
  replace window 1 looking for "Gorge" replacing with "George"
end tell
```

résultat: 2

Toutes les occurences, à savoir deux, de Gorge ont bien été changées par George. Malheureusement, il peut se faire que, dans notre texte, le dit George puisse se gratter la gorge en quelque page de notre livre. Pour éviter un remplacement malencontreux, le script suivant sera préférable:

```
tell application "Tex-Edit Plus F"  
  replace window 1 looking for "Gorge" replacing with "George"with cases matching  
end tell
```

ici une prise en compte de la sensibilité aux caractères en majuscule et minuscule est faite par l'opération de recherche et remplacement. Aussi, si Gorge sera bien modifié en George, gorge sera laissé en l'état.

Pour ajouter d'autres contraintes avec les autres paramètres **whole words matchings** et **global replacing** , il vous suffit de continuer la ligne de script avec **and** :

```
tell application "Tex-Edit Plus F"  
  replace window 1 looking for "Gorge" replacing with "George"with cases matching and global  
  replacing  
end tell
```

Les trois paramètres **cases matching**, **whole words matching**, et **global replacing** peuvent être dans n'importe quel ordre, sous réserve que le premier soit précédé du mot **with**, les suivants étant ajoutés avec le mot de jonction **and** .

Les paramètres **looking for styles** et **replacing with styles** vous permette de rechercher et remplacer des éléments de texte possédant une police de caractères, une taille, un style et une couleur particulière. Ces paramètres nécessitent un enregistrement ("record") comme variable. Dans notre exemple, si nous voulons rechercher toutes les occurrences de George et les changer en style souligné, avec une police de caractère Helvetica de taille 18 et de couleur bleue, alors nous écrivons le script suivant:

```
tell application "Tex-Edit Plus F"  
  replace window 1 looking for "George" replacing with "George" replacing with styles –  
  {font:"Helvetica", style:{underline}, size:18, color:blue}  
end tell
```

Si vous voulez rechercher tous les termes George possédant des attributs de style particuliers et modifier ces attributs, alors écrivez, en substance, ce qui suit:

```
tell application "Tex-Edit Plus F"  
  replace window 1 looking for "George" replacing with "George" looking for styles  
  {font:"Verdana", style:{bold}, size:9, color:black} replacing with styles {font:"Helvetica",  
  style:{underline}, size:18, color:blue}  
end tell
```

Notez que **looking for styles** correspond toujours au texte que l'on cherche et **replacing with styles** correspond toujours au texte de remplacement.

La seconde partie de la commande **replace** invoque des paramètres de remplacement de caractères dédiés au formatage de texte. Le paramètre

replacing multiple spaces with vous permet de remplacer des suites d'espaces utilisées pour un formatage de texte quelconque. Vous pouvez utiliser les valeurs **single space**, **single tab**, **cr** (retour à la ligne), ou **nothing** pour faire le remplacement adéquat. Le script suivant illustre cette procédure, consistant ici à remplacer toutes suites de plus de deux espaces en un simple espace.

```
tell application "Tex-Edit Plus F"  
  replace window 1 replacing multiple spaces with single space  
end tell
```

Maintenant, si vous n'aimez pas la présence de tabulations dans les documents que vous recevez, par exemple si vous sauvez vos courriers dans HyperCard qui ne reconnaît pas les tabulations, utilisez le paramètre **replacing tabs with spaces** suivi du nombre d'espaces (de 0 à 99) que vous désirez affecter aux tabulations.

```
tell application "Tex-Edit Plus F"  
  replace window 1 replacing tabs with spaces 4  
end tell
```

Revert

Annule les modifications apportées à un document édité depuis sa dernière sauvegarde. Ne confondez pas cette commande avec la commande "undo" qui n'annule que la dernière action effectuée.

```
tell window 1 of application "Tex-Edit Plus F"  
  revert  
end tell
```

```
tell application "Tex-Edit Plus F"  
  revert window 1  
end tell
```

Save

Sauve la ou les fenêtres spécifiées sur le disque.

```
try  
  set leFichier to new file  
  open for access leFichier -- voir "open for access"  
  tell application "Tex-Edit Plus F"  
    save window 1 in leFichier with creator "TBB6"  
  end tell  
  close access leFichier -- voir "close access"  
on error msg number num  
  if num = -128 then  
    --annulation de création de fichier -- voir "new file"  
  end if  
  close access leFichier  
end try
```

Search

Cherche une occurrence d'un texte donné dans un autre texte.

L'exemple le plus simple:

```
tell application "Tex-Edit Plus F"  
  search window 1 looking for "day"  
end tell
```

résultat: true

Une autre formulation de ce simple exemple:

```
tell window 1 of application "Tex-Edit Plus F"  
  search looking for "day"  
end tell
```

résultat: true

La recherche s'avérant fructueuse le premier terme "day" trouvé est alors sélectionné dans le texte.

Vous pouvez utiliser les paramètres **cases matching**, **whole words matching**, **searching from cursor**, **global searching**, et **looking for styles** qui sont discutés dans la commande "replace".

Nous illustrerons l'emploi du paramètre **finding** par un simple exemple:

```
tell window 1 of application "Tex-Edit Plus F"  
  search looking for "day" finding next  
end tell
```

A chaque nouvelle exécution du script l'occurrence suivante du terme "day" est cherché. En cas de succès, la commande renvoie la valeur *true* sinon la valeur *false*, ce qui signifie aussi un retour au début du texte du document pour une nouvelle recherche.

Il va s'en dire qu'au moins une occurrence du terme cherché a été déjà trouvée dans le texte pour qu'il soit pertinent de chercher l'occurrence suivante de ce terme (si nous utilisons le terme **next**) ou précédente (si nous utilisons le terme **previous**).

Suivent quelques variations autour du thème de la recherche:

Recherche systématique dans tous les documents ouverts:

```
tell window 1 of application "Tex-Edit Plus F"  
  search looking for "day"with global searching  
end tell
```

Sensibilité de la recherche à la présence des majuscules et minuscules dans le texte.

```
tell window 1 of application "Tex-Edit Plus F"  
  search looking for "Day"with cases matching  
end tell
```

Rechercher la première occurrence du terme "day" dans toutes les fenêtres, sous réserve que ce terme soit un mot et non une partie de mot:

```
tell window 1 of application "Tex-Edit Plus F"  
  search looking for "day"with whole words matching and global searching  
end tell
```

Rechercher, dans toutes les fenêtres, la première occurrence du mot "day" possédant des attributs de style particuliers:

```
tell window 1 of application "Tex-Edit Plus F"  
  search looking for –  
  "day" looking for styles {font:"Verdana", style:{}, size:9} –  
  with whole words matching and global searching  
end tell
```

Nous savons que cette commande retourne la valeur *true* ou *false* en fonction des résultats de la recherche. Utilisons la valeur de ce résultat pour rendre plus intéressante l'emploi de cette commande.

```
tell window 1 of application "Tex-Edit Plus F"  
  set leTerme to "day"  
  search looking for leTerme  
  if the result is true then  
    display dialog "J'ai trouvé " & leTerme & " !" -- voir "display dialog"  
  end if  
end tell
```

Select

Fait une sélection.

```
tell window 1 of application "Tex-Edit Plus F"  
  select ( words 2through 5 )  
end tell
```

Set

Affecte une valeur à une propriété ou des données à un ou plusieurs objets.

```
tell window 1 of application "Tex-Edit Plus F"  
  set color of word 1 to green  
end tell
```

```
tell window 1 of application "Tex-Edit Plus F"  
  set word 2 to "Albert"  
end tell
```

Shift

Indente/désindente le texte spécifié en ajoutant/supprimant des tabulations de début de ligne.

Le script suivant indente, soit la ligne où se trouve le curseur, soit toutes les lignes sélectionnées (souvenez-vous qu'une ligne est une portion de texte terminée par le caractère **return** (passage à la ligne suivante)).

```
tell application "Tex-Edit Plus F"  
  shift selection toward right  
end tell
```

Pour supprimer l'indentation effectuée dans l'exemple précédent:

```
tell application "Tex-Edit Plus F"  
  shift selection toward left  
end tell
```

Avec cette dernière instruction, rien ne se passe si aucune indentation n'existait préalablement.

Smarten

Transforme des caractères qui peuvent apparaître comme normaux sur une autre plateforme que le Mac en caractères typographiques reconnus du Mac.

```
tell window 1 of application "Tex-Edit Plus F"  
  smarten converting quotes  
end tell
```

Les autres paramètres peuvent être utilisés, mais dans des instructions séparées tel que le montre cet exemple:

```
tell window 1 of application "Tex-Edit Plus F"  
  smarten converting quotes  
  smarten converting ligatures  
  smarten converting dashes  
end tell
```

Ces paramètres ne peuvent être joints sur une même ligne de code.

Speak

Lit à voix haute un objet donné, en utilisant pour ce faire le **Speech Manager** .

```
tell window 1 of application "Tex-Edit Plus F"  
  speak  
end tell
```

```
tell application "Tex-Edit Plus F"  
  speak "i like apple script"  
end tell
```

Strip

Supprime divers caractères d'un texte. Soit nous sommes en présence de caractères ASCII non reconnus par le Mac, soit nous désirons supprimer des espaces en trop, soit encore nous désirons convertir des caractères diacritiques, qui sont des caractères utilisés par les langues non-anglophones. Ce script élimine les caractères ASCII non reconnus par le Mac:

```
tell window 1 of application "Tex-Edit Plus F"  
strip removing low ascii  
end tell
```

ou

```
tell application "Tex-Edit Plus F"  
strip window 1 removing low ascii  
end tell
```

Tout comme pour la commande "smarten", vous devez entrer les paramètres individuellement dans les instructions:

```
tell application "Tex-Edit Plus F"  
strip window 1 removing low ascii  
strip window 1 removing high ascii  
strip window 1 removing double spaces  
end tell
```

... parceque Tex-Edit ne peut faire qu'une seule modification à chaque fois.

Strip Line Endings

Supprime les caractères communément utilisés pour signifier une fin de ligne. Les caractères return (CR) de retour à la ligne et les caractères de saut de ligne (CRLF) sont souvent utilisées pour des formatages de textes quelconques sur les plateformes autres que le Mac. Il s'ensuit des indentations erratiques et des lignes de textes anormalement courtes. De même les caractères de saut de ligne apparaissent sous la forme de petits losanges incongrus sur le Mac. Cette commande élimine ces caractères et restaure l'état normal des lignes.

Cette commande ignore le double retour à la ligne et le retour à la ligne associé à une tabulation ou à un espace, ceci afin de préserver un formatage voulu des paragraphes. Néanmoins, ces caractères peuvent être éliminés en premier lieu avec la commande "strip".

Les caractères retour (CR) et retour avec saut de ligne (CRLF) sont souvent utilisés comme caractères de formatage de texte sur les serveur internet et BBS non-Mac. Il s'ensuit souvent qu'apparaissent alors sur un Macintosh des indentations erratiques, des lignes de texte écourtées, et des petits losanges représentant les caractères de retour à la ligne. Cette commande élimine les caractères et restaure la mise-en-page originale.

Ainsi qu'il est mentionné dans Le Manuel, cette commande ignore les caractères retour doublés, les caractères retour associés à une tabulation et les caractères retour associés à un espace. Ceci, dans le souci de préserver le formatage voulu de paragraphe. Néanmoins, si besoin est, ces caractères peuvent être supprimés avec l'emploi préalable de la commande "strip". Ce script élimine d'un document tous les caractères CR utilisés comme fin de ligne:

```
tell window 1 of application "Tex-Edit Plus F"  
strip line endings removing cr  
end tell
```

Pour éliminer d'un document tous les caractères CRLF:

```
tell window 1 of application "Tex-Edit Plus F"  
strip line endings removing crlf  
end tell
```

L'autre paramètre que vous pouvez vouloir utiliser est **with line length**. Ce paramètre vous assure que les seules lignes d'une longueur supérieure à un nombre donné, compris entre 0 et 80, seront soumises à l'élimination de leur caractère de terminaison. Par exemple, si une ligne de 40 caractères est suffisante pour vous mais que le texte que vous manipulez possède des lignes dont les longueurs peuvent être supérieures à 60 caractères, alors vous pouvez affecter à ce paramètre la valeur 40. Dès lors, toute ligne possédant plus de 40 caractères sera éliminée, ainsi que le montre cet exemple:

```
tell window 1 of application "Tex-Edit Plus F"  
strip line endings removing cr with line length 40  
end tell
```

Si vous n'affectez pas de valeur à ce paramètre, la longueur de ligne définie par défaut sera prise en compte.

Sort

Tri les paragraphes sélectionnés. Cette commande a été introduite avec la version 2.9 de Tex-Edit. C'est la commande que vous devez utiliser pour contrôler par script, en mieux, l'équivalent de l'élément de menu **tri** du **Menu Spécial**, qui vous permet un tri alpha-numérique des paragraphes se trouvant dans un texte sélectionné ou un document.

Souvenons-nous qu'un paragraphe est une partie de texte bornée par un caractère return (retour à la ligne).

Le script de tri le plus simple, semblable à l'emploi du menu de tri:

```
tell application "Tex-Edit Plus F"  
sort selection  
end tell
```

A présent, un script impossible à faire en utilisant le menu de tri:

```
tell application "Tex-Edit Plus F"  
  sort contentsof window 1  
end tell
```

Pour faire un tri décroissant:

```
tell application "Tex-Edit Plus F"  
  sort contentsof window 1 with reverse sorting  
end tell
```

Les autres paramètres peuvent être ajoutés à la ligne d'instruction avec les mots réservés **with** ou **without** et **and** :

```
tell application "Tex-Edit Plus F"  
  sort contentsof window 1 with reverse sorting and cases ignored  
end tell
```

Stupefy

Transforme des caractères typographiques en caractères normaux. C'est la commande utilisée pour modifier les caractères typiques du Mac de telle manière que l'on puisse lire correctement le texte modifié sur une autre plateforme.

Pour renverser l'opération faite par cette commande, utilisez la commande "smarten".

L'exemple le plus simple:

```
tell window 1 of application "Tex-Edit Plus F"  
  stupefy converting quotes  
end tell
```

Les autres paramètres peuvent être utilisés dans un même script, mais doivent être associés individuellement à la commande, sans être joints, ce qui nécessite de multiplier les lignes de code.

```
tell window 1 of application "Tex-Edit Plus F"  
  stupefy converting quotes  
  stupefy converting ligatures  
  stupefy converting dashes  
end tell
```

Undo

Annule la plus récente action pouvant être annulée.

```
tell window 1 of application "Tex-Edit Plus F"  
  undo  
end tell
```

ou

```
tell application "Tex-Edit Plus F"  
  undo window 1  
end tell
```